Celestial Navigation

with

Sextant and Calculator HP-42S



Ackermans Harry **2017**

Abbreviations and symbols

$LOP1$ = Line of position 1

$LOP2$ = Line of position 2

$COP1$ = Circle of position 1

$COP2$ = Circle of position 2

$H1$ = Observed altitude (Ho) in $COP1$

$H2$ = Observed altitude (Ho) in $COP2$

Dec = Declination

GHA = Greenwich Hour Angle

$D1$ = Dec of the sun in $COP1$

$G1$ = GHA of the sun in $COP1$

$D2$ = Dec of the sun in $COP2$

$G2$ = GHA of the sun in $COP2$

φ = Latitude of the FIX in the calculations of the COP

L = Longitude of the FIX in the calculations of the COP

$\Delta\varphi$ = Displacement in latitude of the observer

ΔL = Displacement in longitude of the observer

lat_M = Average latitude

$latx$ = Dead-reckoning or assumed observer's latitude

$lonx$ = Dead-reckoning or assumed observer's longitude

FIX = Calculated position

laf = FIX latitude

lof = FIX longitude

$IN1, IN2$ = Intercept of LOP1, LOP2

$AZ1, AZ2$ = Azimuth of LOP1, LOP2

Ha = Apparent altitude

Dog = Distance over Ground of the observer

Cog = Course over Ground of the observer

Ie = Index error of the sextant

$Corr$ = Correction of the sextant height according to its certificate

HoE = Height of eye of the observer

$Da.Mo$ = Day and Month of measurement

$Year$ = Year of measurement

$U.T$ = Universal Time of measurement

$ALTs$ = Measured sextant height

$Limb$ = Top edge, bottom edge or center line of the sun

$Npos$ = Northern intersection of the circles of equal altitude

$Spos$ = Southern intersection of the circles of equal altitude

$Tran$ = Noon latitude of the observer (Transit)

Foreword

The astronomical positioning by the simultaneous measurement of a number of stars leads to a fairly accurate position, but the exercise necessary to identify those stars and to determine their height above the horizon at twilight will often be absent. That is why we limit ourselves to determine our position with the help of the sun.
Using nautical almanacs such as the Macmillan and the HO249 tables, we can traditionally determine our position on the basis of sun's altitudes.
With the advent of the programmable calculators, the entire search, calculation and plot work that follows could be taken over from us. *GHA* and *Dec* of the sun can be calculated with a built-in algorithm, as well as the azimuth and intercept of the lines of position. The somewhat difficult chart work is also digitally reproduced with the final result our position in latitude and longitude *(FIX).*

HP-42S

The HP-42S RPN Scientific is a programmable RPN calculator launched by HP in 1988. Its production was discontinued in 1995 but it is still regarded today as one of the best ever made in terms of quality and ease of programming. Its popularity was so great that he is offered on the site of Thomas Okken under the name "**Free42**: An HP-42S Calculator Simulator" as freeware for many operating systems of *computers and smartphones*. The Free42 computer application is delivered as a zip file in decimal and binary version but it is recommended to install Free42Decimal.exe. Android and iOS applications are in decimal version.
The Free42 app has the advantage that the programs can be saved and loaded as files, backups can be made and loaded into other computers. Instructions can be found here. Moreover, the calculation speed is many times higher than with the hardware version of the HP-42S.
Free42 programs are saved with the file extension ".raw".
In the astronavigation programs described below, only a limited number of keys is used, so that no thorough knowledge of the calculator is required. Those who want to deepen further can go here.

Programs

This text describes two programs for Free42 and HP-42S to navigate by the sun by means of a sextant or theodolite:

ASTRO_TSO.raw: calculates the position *(FIX)* using the azimuth-intercept method of Marcq St. Hilaire. This method, developed in 1875, gradually became the standard procedure for astronomical positioning and is based on the choice of an assumed position (A.P), as close as possible to the true position or the estimated position, and the calculation of the intersection of two so-called *lines of position* (LOP further in the text). This program makes the use of plotting sheets superfluous.

ASTRA_TSO.raw: calculates the position *(FIX)* using an alternative method that directly calculates the intersections of two *circles of position* (COP further in the text). For a *stationary* observer, contrary to the previous method, no assumed position (A.P) is required so that one can determine his position without any idea of where one is. In the case of a *moving observer*, as in the previous program, the latitude of the observer, $latx$, must be known for the calculation of the displacement in longitude ΔL.
Both programs have a built-in almanac of the sun with a maximum deviation of ± 0.5' and an average deviation of ± 0.3' for *GHA* and *Dec*. For verification, the almanac data on the sites backbearing.com and siranah.de can be consulted.

Corrections to the measured sextant height $\boldsymbol{ALTs}$

A number of corrections are applied to the measured sextant height in both programs;

- *Instrument correction:* if the sextant is provided with a certificate, this correction can be taken into account for with $Corr$ (in seconds). If not, the input is = 0
- *Index correction:* is a correction for the index error Ie of the sextant which in principle has to be measured before each new observation. Ie is negative if "off the arc" and positive "on the arc".
- *Dip:* this is a correction that is function of the height of eye HoE. When using an artificial horizon $HoE = 0$.

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

- *Refraction:* this is a correction for the refraction of light according to Bennett's formula.

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

- *Parallax:* as the horizontal parallax of the sun is very small (maximum ± 0.15'), no correction for parallax is applied.
- *Semi diameter (S.D):* this correction is made if the bottom or top edge *(Limb)* of the sun is measured. Between summer and winter the *S.D* varies between ± 15.75' and 16.29', and in the programs the average 16'= 0.2667° is used. For greater accuracy, the value in the program line #331 in ASTRO and #287 in ASTRA must be replaced by the actual value (in degrees!) of the sun-almanac.

Sextant heights

$Ha = ALTs - Ie - Dip$ Apparent altitude

$Ho = Ha - Refraction - S.D$ Observed altitude

$HoE = 0$ implies the use of an artificial horizon

When using an artificial horizon $Dip = 0$ and $Ha = (ALTs - Ie)/2$

Used formulas

- If the Cog and Dog are known then the observer's displacement in latitude and longitude are respectively:

$$\Delta\varphi[°] = Dog[nm]\ .\ \cos Cog\ /60$$

$$\Delta L[°] = \frac{Dog[nm]\ .\ \sin Cog}{\cos lat_M\ .\ 60}$$

in which the average latitude

$$lat_M = latx + \Delta\varphi/2$$

and $latx$ is the latitude before displacement.

- In *ASTRO_TSO.raw* (azimuth-intercept method) the *FIX*-coordinates are

$$laf = \frac{IN2\ .\ \sin AZ1 - IN1\ .\ \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1\ .\ \cos AZ2 - IN2\ .\ \cos AZ1}{\sin(AZ1 - AZ2)\ .\ \cos latx} + lonx$$

where $latx$ and $lonx$ are the coordinates of the A.P after displacement. These values are calculated from the A.P before displacement, Cog and Dog.

Format and units of the input data

latx, lonx		+ (N and E) – (S and W)
Ie	x.xx [minutes]	– OFF the arc + ON the arc
HoE	xx.xx [meter]	Artificial horizon HoE=0
Da.Mo	xx.xx [day.month]	Ex: 5.07 23.12
U.T	xx.xx xx [h.min sec]	Ex: 17.0533 6.0806
latx, lonx, ALTs	xxx.xx x [ddd°mm.m’]	Ex: –15.416 156.093
Limb	Lower=–1 Center=0 Upper=1	
Dog	xxx.x [nautical miles]	
Cog	xxx.x [degrees]	
Corr	xx [seconds]	Ex: –17 12

Program components

ASTRO_TSO.raw: consists of 3 subroutines; *LOP1, LOP2*, and *TRAN*

- *LOP1 (Line of position1):* here the assumed position (A.P), which is usually the dead-reckoning position, and the data of the first observation are entered. The subroutine calculates the intercept *IN1* and azimuth *AZ1* of the first line of position *LOP1* and stores them in their variables with the same name.

- *LOP2 (Line of position2):* on the basis of the *Dog* and *Cog* a new A.P is defined. Together with the data of the second observation, the *IN2* and *AZ2* of *LOP2* are calculated relative to this new A.P. The coordinates of the intersection of the "offset" *LOP1* and *LOP2*, being the *FIX*, are obtained with the above formulas. From a geometric point of view this actually amounts to the translation of *LOP1* in the direction (= *Cog*) of the new A.P over a distance = *Dog*. The coordinates of the *FIX* are stored as "laf"and "lof".
  Since the intercept method ignores the curvatures of the actual position lines, the obtained *FIX* is not our exact position but rather an improved position (compared to AP). The residual error remains tolerable as long as the radii of the circles of equal altitude are not too small and AP is not too far from our actual position. The geometric error inherent to the intercept method can be decreased by iteration, i.e., substituting the obtained *FIX* for AP and repeating the calculations with the same altitudes and GP's (see appendix). This will result in a more accurate position. If necessary, we can reiterate the procedure until the obtained position remains virtually constant (rarely needed).

- *TRAN (Transit):* this part calculates the noon latitude of our position. The input of the estimated longitude $lonx$ only results in a rough estimate of the time of transit because it is calculated on the basis of a fixed transit time of the Greenwich meridian (12h U.T). It is more accurate to start from the actual "Greenwich meridian transit time" and to add 4 minutes per degree of western length or subtract 4 minutes per degree of eastern length.
  "South" and "North" indicate the position of the sun relative to the observer.
  The noon latitude is stored as "laf".

Notes

- *LOP1* and *LOP2* must be entered chronologically

- when using a theodolite, *HoE = 0* should be set, in analogy with an artificial horizon. To prevent *Ha* from being divided by 2, set *HoE* = 0.0001 and the *Dip* correction applied by this is negligible.

ASTRA_TSO.raw : consists of 3 subroutines; *COP1, COP2,* and *TRAN*

- *COP1 (Circle of position1):* $latx$ is the estimated or known latitude of the first observation and will be used in *COP2* to calculate the displacement in longitude ΔL. For a stationary observation (*Dog = 0*) this value is irrelevant.
  After inputting the other measured values, this routine calculates the parameters that define the *Circle of position1*, i.e. *GHA (= G1), Dec (= D1)* and *Ho (= H1).*
- *COP2 (Circle of position2):* with Dog and Cog we calculate the displacement of the first *circle of position COP1*, being $\Delta\varphi$ in latitude and ΔL in longitude. With the other measured values, this routine calculates the parameters that define the *Circle of position2*, i.e. *GHA (= G2), Dec (= D2), Ho (= H2)*, and ultimately our position *(FIX).*

  These 2 circles normally have 2 intersections on the globe. These are our 2 possible positions. When calculating the coordinates of these points of intersection, the choice is made between the northern intersection point ($Npos$) and the southern intersection point ($Spos$). The coordinates of the *FIX* are stored as "laf" and "lof".
  The algorithm for calculating the intersections is discussed in detail in the appendix.

- *TRAN (Transit):* see *ASTRO_TSO.raw*

Notes: see *ASTRO_TSO.raw*

ASTRO* versus *ASTRA

- If one is completely in the dark about his position one can not directly enter a suitable A.P into ASTRO. In case of a *stationary measurement (Dog=0)*, ASTRA does not need this data. With 2 observations, preferably with a difference in azimuth between 30 ° and 150 °, an accurate position can be determined directly and apart from the normal measurement errors



- The iteration process of the ASTRA-program runs very fast in Free42 but can take a lot of time with the HP-42S under certain circumstances (H1≈H2). due to its low clock speed.

- ASTRA always iterates to the most accurate position. With ASTRO, at a great distance between the A.P and the true position and / or at large sextant heights, at least 1 iteration may be required with the calculated FIX as the new A.P.

Free42 and HP-42S commands

- Press the XEQ button. The list of programs that have been loaded appears in the display.

- Press the key below the name of the selected program and after viewing the label, press R/S. The main menu appears in the display, consisting of the 3 subroutines and the EXIT routine to close the program.



- Each routine is started with the key under its name. The requested data is entered via the number pad. Pay particular attention to the units and the format. For negative numbers, first enter the number followed by +/-. An incorrect input can be deleted with the ← key and with RTN you return to the main menu. Each input is followed by R/S.
In a selection menu, press the key below the desired choice. If the program stops with relevant information in the display, press R/S to continue.

- Entered data and calculated values are stored as variables and can be called up with RCL and scroll keys ▲ and ▼

- Calculated values *GHA, Dec, laf* and *lof* are stored in the format [ddd°mm.m']. *AZ1* and *IN1* in [°].

- The results of the observations can be printed out. The print function is activated by PRINT ▲ PON and disabled by PRINT ▲ POFF. The print is available for almost all operating systems via the File menu. With the iPhone the screen must be touched just below the status bar.

Appendix: calculation of the intersections of 2 circles of position

We apply the formulas of the nautical triangle to 2 observations;

$$\sin H1 = \sin D1 \,.\sin(\varphi - \Delta\varphi) + \cos D1 \,.\cos(\varphi - \Delta\varphi)\,.\cos(G1 + L - \Delta L) \qquad (1)$$

$$\sin H2 = \sin D2 \,.\sin\varphi + \cos D2 \,.\cos\varphi \,.\cos(G2 + L) \qquad (2)$$

(1) and (2) are the equations of 2 circles of position in which φ and L are the geographical coordinates of the position of the observer at the time of the second measurement, i.e. the *FIX*. $H1$ and $H2$ are the true sextant heights (after corrections) and $\Delta\varphi$ and ΔL are respectively the displacement in latitude and longitude after the first measurement.

The system of 2 equations with 2 unknowns (φ and L) can not be solved analytically but by iteration. There are 2 conditions attached to this process:
- the iteration must be started in the equation of the circle with the smallest sextant height H, i.e. the circle with the largest radius.
- since for 1 value of φ, 2 values of L are possible and vice versa the appropriate choice must be made here.

If *H2 ≤ H1*

then the starting value of φ (= D2) is placed in equation (2). This gives 2 possible L-values: $L = Acos(\ldots) - G2$ and $L = -Acos(\ldots) - G2$. One chooses the most eastern (= positive) value and since Acos is always positive

$$L = Acos\left(\frac{\sin H2 - \sin D2 \,.\, \sin\varphi}{\cos D2 \,.\cos\varphi}\right) - G2$$

This L-value is placed in (1) and the new φ-value for (2) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1 \ .\cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1 \ .\cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

with $+\sqrt{\ldots}$ for the northern intersection and $-\sqrt{\ldots}$ for the southern intersection.

If *H2 > H1*

then the starting value of φ (= D2) is placed in equation (1). This gives 2 possible L -values: $L = Acos(\ldots) - G1 + \Delta L$ and $L = -Acos(\ldots) - G1 + \Delta L$. One chooses the most western (= negative) value and since Acos is always positive

$$L = -Acos\left(\frac{\sin H1 - \sin D1 \ . \ \sin(\varphi - \Delta\varphi)}{\cos D1 \ .\cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

This L-value is placed in (2) and the new φ-value for (1) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2 \ .\cos^2(G2 + L)}}{\sin H2 + \cos D2 \ .\cos(G2 + L)}\right)$$

with $+\sqrt{\ldots}$ for the northern intersection and $-\sqrt{\ldots}$ for the southern intersection.

In both cases the value of φ is calculated by substitution, by means of the so-called T-formulas, with variables $\tan((\varphi - \Delta\varphi)/2)$ and $\tan(\varphi/2)$ respectively.

The iterations are stopped when 2 consecutive values of φ differ by less than 0.001°

In the choices made in the iterative process described above, it is implicitly assumed that the geographic position (G.P) of the sun at the second observation lies to the west of the G.P of the first observation.

Appendix: iterations in ASTRO

A "running" fix is less accurate than a stationary fix. After all, the *Cog* and *Dog* can only be estimated because the effects of current and wind (drift) are not exactly known in most cases. For a stationary measurement *(Dog = 0)*, the found position can be refined by inputting the *FIX* as the new A.P with the following key sequence in the main menu:

LOP1 RCL LAF R/S RCL LOF R/S and repeating the calculations with the initial input parameters.

Interesting websites

http://thomasokken.nl/free42/
https://www.celnav.de/page2.htm
http://www.teacupnavigation.net/CN.html
http://www.backbearing.com/almanac.html
http://www.siranah.de/html/fr_sail.htm

List of equations used in the text

$$\sin H1 = \sin D1 \,.\, \sin(\varphi - \Delta\varphi) + \cos D1 \,.\, \cos(\varphi - \Delta\varphi) \,.\, \cos(G1 + L - \Delta L) \quad (1)$$

$$\sin H2 = \sin D2 \,.\, \sin\varphi + \cos D2 \,.\, \cos\varphi \,.\, \cos(G2 + L) \quad (2)$$

start φ=D2 in (2)

start φ=D2 in (1)

$$L = Acos\left(\frac{\sin H2 - \sin D2 \,.\, \sin\varphi}{\cos D2 \,.\, \cos\varphi}\right) - G2$$

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1 \,.\, \cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1 \,.\, \cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

$$L = -Acos\left(\frac{\sin H1 - \sin D1 \,.\, \sin(\varphi - \Delta\varphi)}{\cos D1 \,.\, \cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2 \,.\, \cos^2(G2 + L)}}{\sin H2 + \cos D2 \,.\, \cos(G2 + L)}\right)$$

$$\Delta\varphi = Dog \,.\, \cos Cog\, /60 \quad \text{(decimal °)}$$

$$\Delta L = \frac{Dog \,.\, \sin Cog}{\cos lat_M \,.\, 60} \quad \text{(decimal °)}$$

$$lat_M = latx + \Delta\varphi/2$$

$$laf = \frac{IN2 \,.\, \sin AZ1 - IN1 \,.\, \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1 \,.\, \cos AZ2 - IN2 \,.\, \cos AZ1}{\sin(AZ1 - AZ2) \,.\, \cos latx} + lonx$$

$$Lon = \pm Acos\left(\frac{\sin Ho - \sin Dec \,.\, \sin Lat}{\cos Dec \,.\, \cos Lat}\right) - GHA$$

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

ASTRONAVIGATION - SUN HP 42S

```
001 LBL "ASTRO"

002 DEG
003 CF 02
004 CF 03
005 CF 29
006 "......□.ASTRO.□"
007 ├ "If.........zon"
008 AVIEW
009 STOP

010 LBL 00

011 "LOP1"
012 KEY 1 XEQ A
013 "LOP2"
014 KEY 2 XEQ B
015 "TRAN"
016 KEY 4 XEQ C
017 "EXIT"
018 KEY 6 GTO J
019 CLA
020 AVIEW
021 MENU
022 STOP
023 GTO 00

024 LBL A

025 XEQ 01
026 FC? 01
027 RTN
028 STO IN1   (°)
029 RCL 04
030 STO AZ1   (°)
031 XEQ D
032 RTN

033 LBL B

034 XEQ 10
035 FIX 01
036 INPUT Dog
037 INPUT Cog
038 COS
039 RCLx Dog
040 60
041 ÷
042 STO 23          =Δφ
043 RCL 18
044 +
045 XEQ 09
046 STO lat*
047 RCL Cog
048 SIN
049 RCLx Dog
050 RCL 18
051 RCL 23
052 2
053 ÷
054 +
055 COS
056 ÷
057 60
058 ÷
059 RCL 17
060 +
061 XEQ 09
062 STO lon*
063 XEQ 02
064 FC? 01
065 RTN
066 XEQ D
067 XEQ 03
068 SF 02
069 XEQ D
070 PRA
071 PROMPT
072 RTN

073 LBL C

074 XEQ 10
075 FIX 03
076 INPUT lon*
077 XEQ 08
078 15
079 ÷
080 12
081 X<>Y
082 -
083 >HMS
084 "Transit.."
085 PRA
086 ├ ".~."
087 AIP
088 ├ ":"
089 FP
090 100
091 x
092 AIP
093 ├ ".U.T"
094 PROMPT
095 XEQ 02
096 FC? 01
097 RTN
098 LBL c

099 RCL 01
100 90
101 RCL- 08
102 "SOUTH"
103 KEY 1 GTO 14
104 "NORTH"
105 KEY 4 GTO 13
106 CLA
107 PROMPT
108 GTO c

109 LBL 13

110 ±

111 LBL 14

112 +
113 XEQ 09
114 STO laf
115 SF 02
116 SF 03
117 XEQ D
118 CF 03
119 PRA
120 PROMPT
121 RTN

122 LBL 01

123 XEQ 10
124 FIX 03
125 INPUT lat*
126 XEQ 08
127 STO 18
128 INPUT lon*
129 XEQ 08
130 STO 17

131 LBL 02

132 FIX 02
133 INPUT Ie
134 INPUT HoE
135 XEQ I
136 INPUT Da.Mo
137 IP
138 STO 00          =JO
139 LAST X
140 FP
141 100
142 x
143 STO 15          =MO
144 RCL IND ST x
145 STO+ 00
146 FIX 00
147 INPUT Year
148 4
149 ÷
150 FP
151 STO 02          =AA
152 X=0?
153 XEQ a
154 FIX 04
155 INPUT U.T
156 >HR
157 24
158 ÷
159 STO+ 00         =J2
160 FIX 03
161 INPUT ALTs
162 FIX 00
163 INPUT Corr
164 INPUT Limb
165 RCL Year
166 2000
167 -
168 365.25
169 x
170 0.5
171 +
172 RCL- 02
173 365250
174 ÷
175 STO 14
176 RCL 00
177 LAST X
178 ÷
179 +
180 STO 01          =T
181 RAD
182 6283.01961
183 x
184 0.043179665
185 -
186 STO 02
187 0.033417
188 RCL 02
189 SIN
190 x
191 +
192 RCL 02
193 2
194 x
```

```
195 SIN
196 0.0003489
197 x
198 +
199 RCL 01
200 0.300052641
201 x
202 4.938242632
203 +
204 +
205 STO 03
206 SIN
207 0.397777
208 x
209 ASIN
210 >DEG
211 STO 01        =DEC
212 XEQ 09
213 STO DEC
214 RCL 03
215 TAN
216 0.917482
217 x
218 ATAN
219 STO 02
220 RCL 03
221 COS
222 X<0?
223 XEQ b
224 RCL 00
225 360.98564735
226 x
227 RCL 14
228 360007.7
229 x
230 100.465
231 +
232 360
233 MOD           =TSO
234 +
235 360
236 ÷
237 FP
238 PI
239 x
240 2
241 x
242 RCL- 02
243 >DEG          =GHA
244 ENTER
245 XEQ 09
246 STO GHA
247 R↓
248 DEG
249 RCL lon*
250 XEQ 08
251 STO 05
252 +
253 STO 02        =LHA
254 RCL lat*
255 XEQ 08
256 STO 06
257 COS
258 RCL 01
259 SIN
260 x
261 RCL 06
262 SIN
263 RCL 01
264 COS
265 x
266 RCL 02
267 COS
268 x
269 -
270 RCL 01
271 COS
272 ±
273 RCL 02
274 SIN
275 x
276 X<>Y
277 >POL McMillan 159
278 X<>Y
279 X<0?
280 XEQ 12
281 STO 04        =AZ
282 RCL 01
283 SIN
284 RCL 06
285 SIN
286 x
287 RCL 01
288 COS
289 RCL 02
290 COS
291 x
292 RCL 06
293 COS
294 x
295 +
296 ASIN
297 STO 02        =Hc
298 RCL ALTs
299 XEQ 08
300 RCL Ie
301 60
302 ÷
303 -
304 RCL Corr
305 3600
306 ÷
307 +
308 RCL HoE
309 X=0?          Artif. Hor
310 GTO 16
311 SQRT
312 0.0293        =DIP
313 x
314 -

315 LBL 15

316 STO 03
317 4.4           =Refrac
318 +
319 7.31
320 X<>Y
321 ÷
322 RCL+  03
323 TAN
324 1/X
325 60
326 ÷
327 RCL 03
328 X<>Y
329 -
330 RCL Limb
331 0.2667        =S.D
332 x
333 -
334 STO 08        =Ho
335 RCL- 02
336 STO 03        =INT
337 RTN

338 LBL 16

339 R↓
340 2
341 ÷
342 GTO 15

343 LBL a

344 1
345 STO 02
346 RCL 15
347 2
348 X>=Y?
349 RTN
350 1
351 STO+ 00
352 RTN

353 LBL b

354 PI
355 STO+ 02
356 RTN

357 LBL 03        =FIX

358 "If.......*.FIX*"
359 AVIEW
360 RCL AZ1
361 SIN
362 RCLx 03
363 RCL 04
364 SIN
365 RCLx IN1
366 -
367 RCL AZ1
368 RCL- 04
369 SIN
370 STO 07
371 ÷
372 RCL+ 06
373 XEQ 09
374 STO laf
375 RCL 04
376 COS
377 RCLx IN1
378 RCL AZ1
379 COS
380 RCLx 03
381 -
382 RCL: 07
383 RCL 06
384 COS
385 ÷
386 RCL+ 05
387 ABS
388 180
389 X<Y?
390 GTO 11
391 LAST X

392 LBL 04

393 XEQ 09
394 STO lof
395 RTN

396 LBL D

397 CLMENU
398 CLA
399 RCL Da.Mo
400 AIP
401 ├"/"
402 FP
403 100
404 x
```

```
AIP
├ "/"
FIX 00
ARCL Year
├ ".□."
RCL U.T
AIP
├ ":"
FP
100
x
AIP
├ ":"
FP
100
x
AIP
FS? 02
GTO d
├ "IfAZ:."
ARCL 04
├ "°.INT:."
RCL 03
60
x
FIX 01
ARCL STx
PRA
PROMPT
RTN

LBL d

CF 02
├ "If"
RCL Iaf
ENTER
XEQ 07
R↓
X<0?
GTO 05
├ ".N.."
LBL e

FS? 03
RTN
RCL Iof
ENTER
XEQ 07
R↓
X<0?
GTO 06
├ ".E"
RTN

LBL 05

├ ".S.."
GTO e

LBL 06

├ ".W"
RTN

LBL 07

ABS
AIP
├ "°"
FP
100
x
FIX 01
RND
ARCL ST x
RTN

LBL 08        =>d.ddd

IP
LAST X
FP
5
x
3
÷
+
RTN

LBL 09  =>dd.mm.m

IP
LAST X
FP
3
x
5
÷
+
RTN

LBL 10

CLMENU
SF 01
"RTN"
KEY 6 GTO H
MENU
RTN

LBL 11

LAST X
ENTER
SIGN
360
x
-
GTO 04

LBL 12

360
+
RTN
LBL H

CF 01
RTN

LBL I

-1
STO 01
30
ST0 02
58
STO 03
89
STO 04
119
STO 05
150
STO 06
180
STO 07
211
STO 08
242
STO 09
272
STO 10
303
STO 11
333
STO 12
RTN

LBL J

CLMENU
EXITALL
CLST
DEG
FIX 04
END
```

Sight Reduction with 2 Lines of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

lat*, lon* = + (N and E)	HP 42S 1316 bytes
- (S and W)	App Free42 1313 bytes
HoE = meter	Artificial Horizon HoE=0
Limb	lower= -1 centre= 0 upper= 1
Format lat*, lon*, ALTs	= xxx.xx x [ddd°.mm.m']
Format U.T	= xx.xx xx [h . min sec]
Format Da.Mo	= xx.xx [Day . Month]
Dog	= xxx.x [nautical miles]
Cog	= xxx.x [°]
Index error Ie	= - OFF the arc + ON the arc
Format Ie	= x.xx [minutes]
Format Corr	= xx [sec]

ASTRONAVIGATION - SUN HP 42S

```
LBL "ASTRA"

DEG
CF 02
CF 03
CF 29
"......□.ASTRA.□"
├ "If.........zon"
AVIEW
STOP

LBL 00

"COP1"
KEY 1 XEQ A
"COP2"
KEY 2 XEQ B
"TRAN"
KEY 4 XEQ C
"EXIT"
KEY 6 GTO J
CLA
AVIEW
MENU
STOP
GTO 00

LBL A

XEQ 01
FC? 01
RTN
STO 20        =SIN H1
RCL 00
STO 21        =GHA1
RCL 01
STO 22        =DEC1
XEQ D
RTN

LBL B

XEQ 10
FIX 01
INPUT Dog
INPUT Cog
COS
RCLx Dog
60
÷
STO 23        =Δφ
RCL Cog
SIN
RCLx Dog
RCL 19
RCL 23
2
÷
+
COS           =Mean Lat
÷
60
÷
STO 24        =ΔL
XEQ 12
FC? 01
RTN
STO 02        =SIN H2
XEQ 02
FC? 01
RTN
RCL 11
XEQ 09
STO laf
RCL 10
ABS
180
X<Y?
GTO 11
LAST X

LBL b

XEQ 09
STO lof
SF 02
XEQ D
PRA
PROMPT
RTN

LBL C         =Transit

XEQ 10
FIX 03
INPUT lon*
XEQ 08
15
÷
12
X<>Y
-
>HMS
"Transit.."
PRA
├ ".~."
AIP
├ ":"
FP
100
x
AIP
├ ".U.T"
PROMPT
XEQ 12
FC? 01
RTN

LBL c

RCL 01
90
RCL- 08
"SOUTH"
KEY 1 GTO 16
"NORTH"
KEY 4 GTO 15
CLA
PROMPT
GTO c

LBL 15

±

LBL 16

+
XEQ 09
STO laf
SF 02
SF 03
XEQ D
CF 03
PRA
PROMPT
RTN

LBL 01

XEQ 10
FIX 03
INPUT lat*    =DRP
XEQ 08
STO 19

LBL 12

FIX 02
INPUT le
INPUT HoE
XEQ I
INPUT Da.Mo
IP
STO 00        =JO
LAST X
FP
100
x
STO 15        =MO
RCL IND ST x
STO+ 00
FIX 00
INPUT Year
4
÷
FP
STO 02        =AA
X=0?
XEQ 13
FIX 04
INPUT U.T
>HR
24
÷
STO+ 00       =J2
FIX 03
INPUT ALTs
RCL Year
2000
-
365.25
x
0.5
+
RCL- 02
365250
÷
STO 14
RCL 00
LAST X
÷
+
STO 01        =T
RAD
6283.01961
x
0.043179665
-
STO 02
0.033417
RCL 02
```

```
192 SIN
193 x
194 +
195 RCL 02
196 2
197 x
198 SIN
199 0.0003489
200 x
201 +
202 RCL 01
203 0.300052641
204 x
205 4.938242632
206 +
207 +
208 STO 03
209 SIN
210 0.397777
211 x
212 ASIN
213 >DEG
214 STO 01        =DEC
215 XEQ 09
216 STO DEC
217 RCL 03
218 TAN
219 0.917482
220 x
221 ATAN
222 STO 02
223 RCL 03
224 COS
225 X<0?
226 XEQ 14
227 RCL 00
228 360.98564735
229 x
230 RCL 14
231 360007.7
232 x
233 100.465
234 +
235 360
236 MOD           =TSO
237 +
238 360
239 ÷
240 FP
241 PI
242 x
243 2
244 x
245 RCL- 02
246 >DEG
247 STO 00        =GHA
248 XEQ 09
249 STO GHA
250 DEG
251 FIX 00
252 INPUT Corr
253 INPUT Limb
254 RCL ALTs
255 XEQ 08
256 RCL Ie
257 60            =Ie
258 ÷
259 -
260 RCL Corr
261 3600
262 ÷
263 +
264 RCL HoE
265 X=0?          Artif.Hor
266 GTO 18
267 SQRT
268 0.0293
269 x             =DIl
270 -

271 LBL 17

272 STO 03
273 4.4           =Refract
274 +
275 7.31
276 X<>Y
277 ÷
278 RCL+  03
279 TAN
280 1/X
281 60
282 ÷
283 RCL 03
284 X<>Y
285 -
286 RCL Limb
287 0.2667        =S.D
288 x
289 -
290 STO 08        =Ho
291 SIN
292 RTN

293 LBL 13

294 1
295 STO 02
296 RCL 15
297 2
298 X>=Y?
299 RTN
300 1
301 STO+ 00
302 RTN

303 LBL 14

304 PI
305 STO+ 02
306 RTN

307 LBL 18

308 R↓
309 2
310 ÷
311 GTO 17

312 LBL 02

313 CF 04
314 "Npos"
315 KEY 1 GTO 22
316 "Spos"
317 KEY 4 GTO 21
318 CLA
319 PROMPT
320 GTO 02

321 LBL 21

322 SF 04

323 LBL 22

324 ".......*.FIX.*"
325 AVIEW
326 RCL 01
327 STO 05        =STARTφ
328 SIN
329 STO 06
330 LAST X
331 COS
332 STO 07        =COS D2
333 RCL 21        =G1
334 RCL- 24
335 STO 04        =G1- ΔL
336 RCL 22
337 SIN
338 STO 08        =SIN D1
339 LAST X
340 COS
341 STO 09        =COS D1
342 RCL 02
343 RCL 20
344 X<Y?
345 GTO 04
346 LBL 03

347 RCL 02        =SIN H2
348 RCL 06        =SIN D2
349 RCL 05
350 SIN
351 x
352 -
353 RCL÷ 07
354 RCL 05
355 COS
356 ÷
357 ACOS
358 RCL- 00
359 CPX?
360 GTO 04
361 STO 10        =L
362 RCL+ 04
363 COS
364 RCLx 09
365 STO 12
366 X↑2
367 RCL 20
368 X↑2
369 -
370 RCL 08
371 X↑2
372 +
373 X<0?
374 GTO 04
375 SQRT
376 FS? 04
377 ±
378 RCL+ 08
379 RCL 20
380 RCL+ 12
381 ÷
382 ATAN
383 2
384 x
385 RCL+ 23
386 STO 11        =φ
387 RCL- 05
388 ABS
389 0.001
390 X>=Y?
391 RTN
392 RCL11
393 STO 05
394 GTO 03

395 LBL 04

396 RCL 20        =SIN H1
397 RCL 08        =SIN D1
398 RCL 05
```

```
399 RCL- 23
400 STO 12
401 SIN
402 x
403 -
404 RCL 12
405 COS
406 ÷
407 RCL÷ 09
408 ACOS
409 ±
410 RCL- 04
411 CPX?
412 GTO 03
413 STO 10        =L
414 RCL+ 00    =G2+L
415 COS
416 RCLx 07
417 STO 12
418 X↑2
419 RCL 02
420 X↑2
421 -
422 RCL 06
423 X↑2
424 +
425 X<0?
426 GTO 03
427 SQRT
428 FS? 04
429 ±
430 RCL+ 06
431 RCL 02
432 RCL+ 12
433 ÷
434 ATAN
435 2
436 x
437 STO 11        =φ
438 RCL- 05
439 ABS
440 0.001
441 X>=Y?
442 RTN
443 RCL11
444 STO 05
445 GTO 04

446 LBL D

447 CLMENU
448 CLA
449 RCL Da.Mo
450 AIP
451 ├"/"
452 FP
453 100
454 x
455 AIP
456 ├"/"
457 FIX 00
458 ARCL Year
459 ├".□."
460 RCL U.T
461 AIP
462 ├":"
463 FP
464 100
465 x
466 AIP
467 ├":"
468 FP
469 100
470 x
471 AIP
472 FS? 02
473 GTO d
474 ├"IfALTs:."
475 FIX 03
476 ARCL ALTs
477 PROMPT
478 RTN

479 LBL d

480 CF 02
481 ├"If"
482 RCL Iaf
483 ENTER
484 XEQ 07
485 R↓
486 X<0?
487 GTO 05
488 ├".N.."

489 LBL e

490 FS? 03
491 RTN
492 RCL lof
493 ENTER
494 XEQ 07
495 R↓
496 X<0?
497 GTO 06
498 ├".E"
499 RTN
500 LBL 05

501 ├".S.."
502 GTO e

503 LBL 06

504 ├".W"
505 RTN

506 LBL 07

507 ABS
508 AIP
509 ├"°"
510 FP
511 100
512 x
513 FIX 01
514 RND
515 ARCL ST x
516 RTN

517 LBL 08      =>d.ddd

518 IP
519 LAST X
520 FP
521 5
522 x
523 3
524 ÷
525 +
526 RTN

527 LBL 09 =>dd.mm.m

528 IP
529 LAST X
530 FP
531 3
532 x
533 5
534 ÷
535 +
536 RTN

537 LBL 10

538 CLMENU
539 SF 01
540 "RTN"
541 KEY 6 GTO H
542 MENU
543 RTN

544 LBL 11

545 LAST X
546 ENTER
547 SIGN
548 360
549 x
550 -
551 GTO b

552 LBL H

553 CF 01
554 RTN

555 LBL I

556 -1
557 STO 01
558 30
559 ST0 02
560 58
561 STO 03
562 89
563 STO 04
564 119
565 STO 05
566 150
567 STO 06
568 180
569 STO 07
570 211
571 STO 08
572 242
573 STO 09
574 272
575 STO 10
576 303
577 STO 11
578 333
579 STO 12
580 RTN

581 LBL J

582 CLMENU
583 EXITALL
584 CLST
585 DEG
586 FIX 04
587 END
```

Sight Reduction with 2 Circles of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

lat*, lon* = + (N and E)	HP 42S 1353 bytes
- (S and W)	App Free42 1348 bytes
HoE = meter	Artificial Horizon HoE=0
Limb	lower= -1 centre= 0 upper= 1
Format lat*, lon*, ALTs	= xxx.xx x [ddd°.mm.m']
Format U.T	= xx.xx xx [h . min sec]
Format Da.Mo	= xx.xx [Day . Month]
Dog	= xxx.x [nautical miles]
Cog	= xxx.x [°]
Index error Ie	= - OFF the arc + ON the arc
Format Ie	= x.xx [minutes]
Format Corr	= xx [sec]